

# GESStabs

---

## HTML-Charts in GESStabs



Gesellschaft für Software  
in der Sozialforschung mbH

Waterloohain 6 - 8  
22769 Hamburg  
Tel.: +49 40 - 853 753 - 0  
Fax: +49 40 - 853 753 - 33  
[www.gessgroup.de](http://www.gessgroup.de)

# Inhaltsverzeichnis

1. Anwendung . . . . .	1
2. Syntaxübersicht . . . . .	9
<b>Stichwortverzeichnis</b>	<b>11</b>

Ab Software-Version 5.1.3.5 bietet GESStabs die Möglichkeit, Tabellen auch in der optionalen Ausgabe in HTML<sup>1</sup> durch einfache Abbildungen zu ergänzen. Hierzu dient die neue Anweisung HTMLCHART.

Solch grafische Darstellungen sollen in der Regel nicht die Datentabelle ersetzen, sondern einen Ausschnitt aus den in der Tabelle präsentierten Ergebnissen zur schnellen Erfassung abbilden. Die klassische Aufgabe von Tabellen ist es, eine große Informationsmenge in übersichtlicher Form zu zeigen. Das Chart soll die Aufmerksamkeit des Lesers auf besonders interessante Einzelheiten lenken. Eine wesentliche Funktion der Skriptsprache zu HTMLChart sind deshalb die Selektionsmechanismen, die zur Auswahl bestimmter Tabellenbereiche sowie spezifischer Darstellungsformen zur Verfügung stehen.

In der von GESStabs generierten HTML-Ausgabe werden die Diagramme innerhalb des Tabellenrahmens oberhalb der Quelltablelle angezeigt. Im Skript werden die Chart-Anweisungen direkt im Anschluss an das TABLE-Statement geschrieben, aus dessen Ergebnistabelle Daten und Texte entnommen werden. Eine Tabelle kann um beliebig viele Diagramme ergänzt werden. Die Steuerung folgt in ihren Grundzügen der Chart-Erstellung in GESStabs Artist und OfficeChart.

## 1. Anwendung

Grundlage der folgenden Anwendungsbeispiele ist eine einfache Tabelle, in der zwei Variablen gegeneinander gekreuzt werden: Hamburger Sehenswürdigkeiten (v2) und eine (fiktive) Bewertung ihrer Beliebtheit (v4). Die fünf-stellige Skala in der Tabellenseite wurde durch die Bildung zweier Overcodes und des Mittelwertes ergänzt. Den Tabellenrahmen bilden eine Total-Spalte sowie eine Basis-Zeile, die Zellen enthalten Spaltenprozente und Absolutwerte bzw. Mittelwerte.

Sehenswürdigkeiten in Hamburg						
Col % Abs.	Total	Hamburg				
		Hafen	Michel	Reeperbahn	Stadtpark	Alster
<b>N</b>	261	33	56	71	64	37
<b>Beliebtheit</b>						
<b>sehr hoch</b>	<b>11%</b> 28	<b>15%</b> 5	<b>11%</b> 6	<b>6%</b> 4	<b>8%</b> 5	<b>22%</b> 8
<b>hoch</b>	<b>31%</b> 80	<b>18%</b> 6	<b>25%</b> 14	<b>34%</b> 24	<b>34%</b> 22	<b>38%</b> 14
<b>mittel</b>	<b>25%</b> 65	<b>18%</b> 6	<b>21%</b> 12	<b>34%</b> 24	<b>27%</b> 17	<b>16%</b> 6
<b>niedrig</b>	<b>23%</b> 59	<b>30%</b> 10	<b>29%</b> 16	<b>18%</b> 13	<b>25%</b> 16	<b>11%</b> 4
<b>sehr niedrig</b>	<b>11%</b> 29	<b>18%</b> 6	<b>14%</b> 8	<b>8%</b> 6	<b>6%</b> 4	<b>14%</b> 5
<b>Top-2</b>	<b>41%</b> 108	<b>33%</b> 11	<b>36%</b> 20	<b>39%</b> 28	<b>42%</b> 27	<b>59%</b> 22
<b>Bottom-2</b>	<b>34%</b> 88	<b>48%</b> 16	<b>43%</b> 24	<b>27%</b> 19	<b>31%</b> 20	<b>24%</b> 9
<b>Mittelwert</b>	<b>2,9</b>	<b>3,2</b>	<b>3,1</b>	<b>2,9</b>	<b>2,9</b>	<b>2,6</b>
GESStabs Gesellschaft für Software in der Sozialforschung mbH						


*Zugrunde liegende Beispieltabelle*

<sup>1</sup>Skriptbefehl zur Ausgabe in HTML: HTML = "<Dateiname>";

## Einfaches HTML-Chart

Im Anschluss an das TABLE-Statement wird das sich darauf beziehende Chart-Statement geschrieben:

```
CELLELEMENTS = COLUMNPERCENT ABSOLUTE;  
FRAMEELEMENTS = ABSROW TOTALCOLUMN;  
TABLE = v2 BY v4 mean(v4);  
  
HTMLCHART =  
| COLUMNS POSITION 1:5  
| ROWS POSITION 1:5  
;
```



*Einfaches Säulendiagramm*

Dies ist die einfachste Form der HTML-Chart-Anweisung: **HTMLCHART** stößt die Erstellung eines Diagramms an. Wird keine Angabe zur gewünschten Diagrammform gemacht, zeichnet GESStabs standardmäßig ein Säulendiagramm (COLUMNS). Per Default werden außerdem die in der Tabelle abgebildeten ColumnPercent-Werte dargestellt. Zur Darstellung werden die ersten fünf Spalten (COLUMNS POSITION 1:5) und Zeilen (ROWS POSITION 1:5) der Tabelle ausgewählt. Die Ausprägungen des Tabellenkopfes werden in der X-Achse der Abbildung gezeigt, die Zellenwerte bestimmen die Höhe der Diagrammsäulen. Die farbliche Gestaltung richtet sich nach einem zuvor definierten Farbschema<sup>2</sup>, dessen Farben den Ausprägungen der Y-Achse in der Reihenfolge ihrer Darstellung zugeordnet werden.

## Definition des Diagrammtyps


Drei Modifikationen mögen die Möglichkeiten der Diagrammgestaltung demonstrieren: Farben, Form und INVERSE:

```
GESSCARTCOLORS = $0066CC $CC3399 $33CCCC $66FF66 $CC00FF;  
HTMLCHART  
INVERSE
```

<sup>2</sup>Siehe nächstes Beispiel und [GESSCartColors](#).

```
FORM BARS =
| COLUMNS POSITION 1:5
| ROWS POSITION 1:5
;
```

GESSCHARTCOLORS definiert ein für die nachfolgenden Skriptteile geltendes Farbschema, via FORM BARS wird ein Balkendiagramm angefordert. INVERSE bewirkt eine Drehung der Darstellung, d.h. die Kategorien der Tabellenseite dienen nun als Beschriftung der (senkrechten) Basislinie, während die GESSChartColors den in der Diagrammlegende stehenden Ausprägungen des Tabellenkopfes zugeordnet werden.




*Balkendiagramm, INVERSE*

## Definition eines Diagrammtitels

Eine Variation des Balkendiagramms ist die OPTION STACKED: Hiermit wird ein gestapeltes Balkendiagramm angefordert, das für die vorliegenden Daten die bevorzugte Darstellungsform ist. INVERSE nehmen wir zurück, beschränken uns auf ausgewählte Datenreihen und ergänzen mit TITLE einen Diagrammtitel.


```
HTMLCHART
FORM BARS
OPTION STACKED
TITLE "Beliebtheit ausgewählter Hamburger Sehenswürdigkeiten"
=
| COLUMNS POSITION 2:4
| ROWS POSITION 1:5
;
```



*Gestapeltes Balkendiagramm*

## Ansprache von CellElements und Labelcodes

Im nächsten Beispiel wird ein Liniendiagramm angefordert, das die Werte der Overcodes (Top-2 und Bottom-2) abbildet:



*Liniendiagramm*

```
HTMLCHART
FORM LINES
CELLELEMENT ABSOLUTE
=
| COLUMNS POSITION 1:5
| ROWS 1/65002 1/65003
;
```

FORM LINES erzeugt ein Liniendiagramm. Diesmal werden die Absolutwerte dargestellt, das CellElement<sup>3</sup> wird mit dem gleichnamigen Schlüsselwort angefordert: CELLELEMENT ABSOLUTE.

<sup>3</sup>sofern in der zugrundeliegenden Tabelle vorhanden


Neu ist auch die Ansprache der Overcodes, die als Codes 65002 und 65003 der ersten Labelliste in der Tabelle (daher: 1/65002 1/65003) selektiert werden. Auf das Schlüsselwort **POSITION** wird in der Zeilenauswahl folglich verzichtet.

Für Liniendiagramme steht optional der Parameter **LINETENSION** <Wert> zur Verfügung, der die Linienspannung steuert. Ein Wert von 0 erzeugt gerade Verbindungslinien zwischen den Datenpunkten, aufsteigend wird die Linienführung kurviger. Sinnvolle Werte liegen zwischen 0 und 0.5<sup>4</sup>, voreingestellt ist eine LineTension von 0.08.

## Ansprache von FrameElements

Die vornehmste Aufgabe eines Diagramms ist die Abbildung selektiver Informationen. Häufig wird nur die Totalspalte/-zeile abgebildet. Diese FrameElements können in der Auswahl darzustellender Tabelleninhalte mit den Schlüsselwörtern **TOTALCOLUMN** und **TOTALROW** angesprochen werden. Im unten stehenden Syntaxbeispiel wird die TotalColumn ausgewählt und zudem eine Variation der Zeilenauswahl gezeigt: Angesprochen werden die ersten fünf Codes der ersten Labelliste<sup>5</sup>. Als Darstellungsform wird mittels **FORM PIE** ein Kreisdiagramm gewählt.

```
HTMLCHART
CELLELEMENT COLUMNPERCENT
FORM PIE
TITLE 'Gesamtzufriedenheit mit Hamburger Sehenswürdigkeiten'
=
| COLUMNS TOTALCOLUMN
| ROWS 1/1:5
;
```



*Kreisdiagramm*


<sup>4</sup>Formal werden auch höhere Werte akzeptiert. **LINETENSION 5** erzeugt zum Beispiel sehr hübsche Linienknäuel.

<sup>5</sup>Zufällig entsprechen die Labelcodes 1 bis 5 in dieser Tabelle den ersten fünf Tabellenzeilen, daher führte **ROWS POSITION 1:5** hier zu einem identischen Ergebnis.

## Legendenposition

Als Spielart des Kreisdiagramms steht für HTML-Charts das Ringchart zur Verfügung (FORM DOUGHNUT). Im unten stehenden Beispiel wird zudem die Legende mittels `LEGENDPOSITION RIGHT` rechts neben dem Diagramm platziert.

```
HTMLCHART
CELLELEMENT ABSOLUTE
LEGENDPOSITION RIGHT
FORM DOUGHNUT
=
| COLUMNS TOTALCOLUMN
| ROWS 1/1:5
;
```



*Ringdiagramm*

Beachte: Die Diagrammtypen `PIE` und `DOUGHNUT` können keine mehrdimensionalen Tabellenmatrizen, sondern ausschließlich einzelne Datenreihen abbilden.

## GESChartColors


Zu guter Letzt soll noch einmal auf die Farbgestaltung der HTML-Charts eingegangen werden. Grundsätzlich wird ein Farbschema mit dem Schlüsselwort `GESCHARTCOLORS` definiert, es folgen die gewünschten Farbcodes. Die Farben werden in der Reihenfolge ihrer Definition den in einem Diagramm abgebildeten Labelcodes zugeordnet. Nun kann es passieren, dass für verschiedene Abbildungen unterschiedliche Farben oder -kombinationen verwendet werden sollen. Dazu kann entweder vor der betreffenden `HTMLCHART`-Anweisung ein komplett neues Farbschema definiert werden (das alte wird für folgende Skriptteile überschrieben), oder es werden nur ausgewählte Positionen des bestehenden Farbschemas ausgetauscht. Letzteres wird mittels `GESCHARTCOLORS <Nummer>` erreicht: `GESCHARTCOLORS 3 = $FFKKDD`; ersetzt beispielsweise nur die dritte Position des zuvor definierten Farbschemas, `GESCHARTCOLORS 3 = $DDDDDD $CCBBAA`; die an dritter und vierter Stelle definierten Farben.

Ein Beispiel:



Für die vorangegangenen Diagramme wurde eine Farbpalette von Grün bis Rot definiert: `GESSHARTCOLORS = $00B050 $92D050 $FFFF00 $FFC000 $FF0000;`. Nun soll die Mittelwerte-Zeile in einem Säulendiagramm dargestellt werden. Für die erste (und einzige) Datenreihe würde der erste Farbwert der definierten `GESSHARTCOLORS` ausgewählt werden. Anstelle eines Grüntons sollen die Säulen blau eingefärbt werden, die erste Position des bestehenden Farbschemas wird ausgetauscht und anschließend das Chart angefordert:

```
GESSHARTCOLORS 1 = $0066CC;
HTMLCHART
=
| COLUMNS POSITION 1:5
| ROWS 2/1
;
```




*Selektive Darstellung der Mittelwerte*

Mit `INVERSE` wird die Darstellungsrichtung nun noch gedreht. In der Folge wechselt die Bezeichnung des Tabellenkopfes in die Diagrammlegende und jede Säule erhält eine eigene Farbe. Das Farbschema soll unverändert mit Blau beginnen und ab der zweiten Stelle neue Farbcodes erhalten<sup>6</sup>:

```
GESSHARTCOLORS 2 = $CC3399 $33CCCC $66FF66 $CC00FF $FF9966;
HTMLCHART
INVERSE
=
| COLUMNS POSITION 1:5
| ROWS 2/1
;
```

<sup>6</sup>Kurze Anmerkung im Zeichen der Effektivität: In diesem Fall hätte es einen `GESSHARTCOLORS`-Befehl eingespart, wäre gleich zu Beginn des Beispiels ein komplett neues Farbschema definiert worden. Zu exemplarischen Zwecken sind kleine Umwege erlaubt.



*Selektive Darstellung der Mittelwerte, INVERSE*

## 2. Syntaxübersicht

Grundstruktur:

**HTMLCHART** {<Optionen>} = <Zellen>;

Optionen:

**<Optionen> ::= [ TITLE <String> | FORM <Form> | OPTION STACKED | CELLELEMENT <Cellelement> ] | [ LEGENDPOSITION <Position> ] | [ LINE-TENSION <Nummer> ] | [ INVERSE ] | WIDTH <Pixel>**

**TITLE <String>**

Diagrammtitel (String ohne Zeilenumbruch)

Ist kein TITLE angegeben, bleibt der Diagrammtitel leer.

**FORM <Form>**

Diagrammtyp

**<Form> ::= [ COLUMNS | BARS | PIE | DOUGHNUT | LINES ]**

**OPTION STACKED**

Stapelung in COLUMNS-, BARS- und LINES-Diagramm

**LINE-TENSION <Nummer>**

Linienspannung

Default: 0.08

**CELLELEMENT <Cellelement>**

Auswahl eines in der Tabelle vorhandenen Zellenelements

Kann bei Tabellen mit nur einem Zellenelement ausgelassen werden.

**LEGENDPOSITION <Position>**

Positionierung der Legende

**<Position> ::= [ LEFT | RIGHT | TOP | BOTTOM ]**

**INVERSE**

Invertierte Darstellung des Tabelleninhalts

**WIDTH <Pixel>**

Breite des Charts in Pixel

Voreinstellung mittels HTMLCHARTWIDTH möglich (siehe unten)

Zellen:

**<Zellen> ::= [ | ROWS <Rows> ] [ | COLUMNS <Columns> ]**

Angabe des darzustellenden Tabellenbereichs

**<Rows> ::= [ POSITION <Nummer> | <Variablennummer>/<Labelcode/s>\*n | TOTALROW ]**

**<Columns> ::= [ POSITION <Nummer> | <Variablennummer>/<Labelcode/s>\*n | TOTALCOLUMN ]**

Zur Auswahl der Bereiche können die (POSITION) Zeilen-/Spaltennummern oder eine Kombination aus Variablennummer und Labelcode/s angegeben werden. Spannbereiche können in üblicher GESStabs-Manier mit Doppelpunkt zwischen zwei Nummern gekennzeichnet werden. Auch die Ansprache von Totalzeile und -spalte ist möglich.

Globale Einstellungen:

**HTMLCHARTWIDTH = <Pixel>;**

Breite des Charts in Pixel

Default: 500

**GESSHARTCOLORS <Position> = <Farbcode/s>;**

Definition eines Farbschemas

Einsatz einer Position bewirkt, dass die Farbe/n in einem zuvor definierten Farbschema ab der beschriebenen Position ausgewechselt werden.

# Stichwortverzeichnis

Balkendiagramm, [3](#)  
BARS, [3](#)  
  
CELLELEMENT, [4](#)  
  
FORM, [3](#)  
FORM DOUGHNUT, [6](#)  
FORM LINES, [4](#)  
FORM PIE, [5](#)  
  
GESSCHARTCOLORS, [3](#), [6](#)  
Gestapeltes Balkendiagramm, [3](#)  
  
HTMLCHART, [2](#)  
  
INVERSE, [3](#)  
  
Kreisdiagramm, [5](#)  
  
Labelcode, [4](#)  
LEGENDPOSITION, [6](#)  
LINETENSION, [5](#)  
Liniendiagramm, [4](#)  
  
OPTION STACKED, [3](#)  
  
Ringdiagramm, [6](#)  
  
Säulendiagramm, [2](#)  
Syntax, [9](#)  
  
TITLE, [3](#)  
TOTALCOLUMN/-ROW, [5](#)